
ABSTRACT: We introduce the Macaulay2 package SpecialFanoFourfolds, a package that provides
functions for working with cubic fourfolds, Gushel–Mukai fourfolds, and some other special Fano
fourfolds.

INTRODUCTION. The package SpecialFanoFourfolds in Macaulay2 [2] provides support for some types
of Hodge-special Fano fourfolds, which can be represented as smooth hypersurfaces of low degree r
in some fixed ambient fivefold V. Roughly speaking, a smooth degree-r hypersurface X ⊂ V is called
Hodge-special if it contains an algebraic surface S whose cohomology class does not come from the
ambient fivefold V. In the parameter space of all smooth hypersurfaces X ⊂ V of degree r , the locus of
Hodge-special fourfolds is called the Noether–Lefschetz locus.

One important case is that of smooth cubic hypersurfaces in P5 (cubic fourfolds for short). The
Noether–Lefschetz locus in the 20-dimensional moduli space C = (P(OP5(3)) \ Disc3

P5)/PGL6 of cubic
fourfolds is a countable union of irreducible hypersurfaces Cd ⊂ C, where d > 6 with d ≡ 0, 2 mod 6.
The hypersurface Cd parametrizes cubic fourfolds of discriminant d , that is, the set of cubic fourfolds X
which contain a surface S such that the discriminant of the saturated lattice spanned by h2

X and [S] in
H 2,2(X, Z) := H 4(X, Z) ∩ H 2(�2

X ) is d (here h X stands for the class of a hyperplane section of X ). For
general results on cubic fourfolds, we refer the reader to [3; 4].

Another important case is that of smooth quadric hypersurfaces in a 5-dimensional linear section
V ⊂ P8 of the cone in P10 over the Grassmannian G(1, 4) ⊂ P9. Such hypersurfaces in V are known
as Gushel–Mukai fourfolds (GM fourfolds for short), and are parametrized by a moduli space GM of
dimension 24. Outside a closed subset of codimension 2 in GM, we have that the ambient fivefold V is
smooth, and hence it is isomorphic to a hyperplane section of G(1, 4) ⊂ P9. In such case, a GM fourfold
X ⊂ V is called ordinary. The Noether–Lefschetz locus in GM is a countable union of hypersurfaces GMd ,
labeled by the possible values of the discriminant d , which are the integers d > 8 with d ≡ 0, 2, 4 mod 8.
If d ≡ 2 mod 8, then GMd is the union of two irreducible components GM′

d ∪ GM′′

d ; otherwise it is
irreducible. For general results on Gushel–Mukai fourfolds, we refer the reader to [1].

One of the most important questions about cubic fourfolds and GM fourfolds is to establish when they
are rational. There are classical examples of such fourfolds which are rational, but it is conjectured that
most of them are not rational.

In the following section, we give just a brief introduction to how to use the package. For more
computational details and recent applications of the package, see [12].

1. THE MAIN FUNCTIONS OF THE PACKAGE.

Constructing Hodge-special fourfolds. The first task of the package is to provide tools to create Hodge-
special fourfolds and locate them in the corresponding Noether–Lefschetz locus. The most basic con-
structor of such objects is given by the function specialFourfold, which typically takes as input a
pair (S, X) of a surface S and a fourfold X with S ⊂ X , and returns an object that can be used anywhere
in place of X . It is possible to specify the ambient fivefold V as third argument. On the returned object
we can apply the function discriminant to find the component of the Noether–Lefschetz locus where
the fourfold belongs. Clearly, this depends on the surface S since the fourfold X could belong to several
components.

In the following example, we take a random cubic fourfold X ⊂ P5 containing a plane S and verify that
it belongs to the components C8 ⊂ C. The value d = 8 comes from the fact that the self-intersection (S)2

X

of S in X is 3, so that the discriminant of X is

d = det
(

h4
X h2

X · S
S · h2

X (S)2
X

)
= det

(
deg X deg S
deg S (S)2

X

)
= det

(
3 1
1 3

)
= 8,

which is computed as

i1 : needsPackage "SpecialFanoFourfolds";

i2 : K = ZZ/65521;

i3 : S = random({3:{1}},0_(PP_K^5)); -- a random plane in PP^5

o3 : ProjectiveVariety, surface in PP^5

i4 : X = specialFourfold(S,random(3,S));

o4 : ProjectiveVariety, cubic fourfold containing a surface of degree 1 and sectional genus 0

i5 : discriminant X

o5 = 8

Note that in the line i4, we could omit the second argument by just typing X = specialFourfold S.
In this case, the cubic fourfold X will be chosen randomly from those that contain S. The same applies in
the case of GM fourfolds.

https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/SpecialFanoFourfolds/html/_special__Fourfold.html
https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/SpecialFanoFourfolds/html/_discriminant_lp__Special__Gushel__Mukai__Fourfold_rp.html

Below we construct an ordinary GM fourfold X as the transverse intersection of G(1, 4) with a
hyperplane and a hyperquadric through a plane S ⊂ G(1, 4) of type σ3,1. We verify that this fourfold
belongs to the component GM′′

10.

i6 : S = schubertCycle({3,1},GG_K(1,4));

o6 : ProjectiveVariety, surface in PPˆ9 (subvariety of codimension 4 in GG(1,4) ⊂ PPˆ9)

i7 : X = specialFourfold S; -- random GM fourfold through S

o7 : ProjectiveVariety, GM fourfold containing a surface of degree 1 and sectional genus 0

i8 : discriminant X

o8 = 10

i9 : describe X

o9 = Special Gushel-Mukai fourfold of discriminant 10(’’)
containing a surface in PP^8 of degree 1 and sectional genus 0
cut out by 6 hypersurfaces of degree 1
and with class in G(1,4) given by s_(3,1)
Type: ordinary

Finally we construct a complete intersection of three quadrics in P7 containing a plane, thought of as a
quadric hypersurface in a complete intersection of two quadrics V ⊂ P7; see also [9]. We only specify
the surface S, and let Macaulay2 choose the fourfold and the ambient fivefold V randomly.

i10 : S = random({5:{1}},0_(PP_K^7)); -- a random plane in PP^7

o10 : ProjectiveVariety, surface in PP^7

i11 : X = specialFourfold S;

o11 : ProjectiveVariety, complete intersection of three quadrics in PP^7
containing a surface of degree 1 and sectional genus 0

i12 : discriminant X

o12 = 31

Count of parameters. A surface S ⊂ V corresponds to a point of some irreducible component S of
the Hilbert scheme Hilb(V). Thus it is useful to have an estimate on the dimension of the family of
fourfolds X ⊂ V containing some surface of S. The function parameterCount applied to a fourfold
X ⊃ S automates a count of parameters based on the following proposition.

Proposition 1.1 ([7; 11; 12]; see also [5; 8]). Let S ⊂ V be a smooth irreducible surface which is contained
in a smooth hypersurface X ⊂ V of degree r . Assume that

(1) h1(NS/V) = 0, and

(2) h1(OS(r)) = 0 and h0(IS/V(r)) = h0(OV(r)) − χ(OS(r)).

Then there is a unique irreducible component S ⊂ Hilb(V) of the Hilbert scheme of V that contains [S],
and the family XS ⊂ P(H 0(OV(r))) of the hypersurfaces in V of degree r containing some surface of the
family S has codimension at most

dim
(
P

(
H 0(OV(r))

))
−

(
h0(NS/V) + h0(IS/V(r)) − h0(NS/X ) − 1

)
. (1-1)

https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/SpecialFanoFourfolds/html/_parameter__Count.html

Particularly important is the case when the value of (1-1) is 1 since in this case we generally obtain
a geometric description of a whole irreducible component of the Noether–Lefschetz locus. See [11]
and [12] for a systematic use of the function parameterCount to deduces some birational properties of
the first components of the Noether–Lefschetz locus in the moduli space of GM fourfolds; see also [7].

In the example below, we execute the function to deduce that the family of cubic fourfolds containing
a plane is the whole component C8.

i13 : X = o4; -- cubic fourfold constructed in the input line i4

o13 : ProjectiveVariety, cubic fourfold containing a surface of
degree 1 and sectional genus 0

i14 : parameterCount(X,Verbose=>true)
-- S: plane in PP^5
-- X: smooth cubic hypersurface in PP^5
-- h^1(N_{S,P^5}) = 0
-- h^0(N_{S,P^5}) = 9
-- h^1(O_S(3)) = 0, and h^0(I_{S,P^5}(3)) = 46 = h^0(O_(P^5)(3)) - \chi(O_S(3));
-- h^0(N_{S,X}) = 0
-- codim{[X] : S ⊂ X} <= 1

o14 = (1, (46, 9, 0))

Establishing rationality of fourfolds. The function detectCongruence may detect eventual congruences
of (re − 1)-secant curves of degree e ≥ 1 to the surface S inside the ambient fivefold V; see [8] for
theoretical details. More precisely, it detects if for some e ≥ 1 there exists a unique curve of degree e
passing through the general point of V, which is contained in V and intersects the surface S at re − 1
points. In most cases, this is accomplished by considering the rational map ϕ : V 99K Z ⊂ PN defined by
the linear system of hypersurfaces of degree r in V through S. Indeed, if ϕ is birational onto its image
Z = ϕ(V), it induces a 1–1 correspondence⋃

e≥1

{
curves of degree e in V passing through a general

point p ∈ V and that are (re − 1)-secant to S

}
≃
−→

{
lines contained in Z ⊂ PN and

passing through ϕ(p)

}
.

So that, one can analyze the pull-backs of the lines in V passing through a general point on Z ; see the
documentation for the functions coneOfLines and RationalMapˆ* ProjectiveVariety. The object
returned looks like a rational map h : V 99K Hilb(V) which takes a point p ∈ V and returns the unique
curve of degree e, (re − 1)-secant to S, and passing through p.

In most interesting cases, the general curve of the congruence can be realized as the general fiber of
the rational map µ : V 99K W = µ(V) defined by the linear system of hypersurfaces of degree re −1 in V

having points of multiplicity e along the surface S (so that necessarily dim W = 4). This map can be
obtained with the command map h, where h is the congruence. The importance of the congruences is
that, under a mild hypothesis of transversality (see [8]), we have that the restriction of the map µ induces
a birational map µ|X : X 99K W . So that X is rational if and only if W is.

In the example below, we show that a GM fourfold X containing a σ3,1-plane

S ⊂ V = P8
∩ G(1, 4) ⊂ P9

https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/SpecialFanoFourfolds/html/_parameter__Count.html
https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/SpecialFanoFourfolds/html/_detect__Congruence.html
https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/MultiprojectiveVarieties/html/_cone__Of__Lines.html
https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/MultiprojectiveVarieties/html/___Multirational__Map_sp%5E_st_st_sp__Multiprojective__Variety.html
https://faculty.math.illinois.edu/Macaulay2/doc/Macaulay2/share/doc/Macaulay2/SpecialFanoFourfolds/html/_map_lp__Congruence__Of__Curves_rp.html

is rational. Indeed S ⊂ V admits a congruence of 1-secant lines from which we get a birational map
from X to a smooth quadric hypersurface W ⊂ P5.

i15 : X = o7; -- GM fourfold constructed in the input line i7

o15 : ProjectiveVariety, GM fourfold containing a surface of degree 1 and sectional genus 0

i16 : (S, V) = (surface X, ambientFivefold X);

i17 : h = detectCongruence X;

o17 : Congruence of 1-secant lines to S in V

i18 : mu = map h;

o18 : RationalMap (dominant rational map from V to hypersurface in PP^5)

i19 : (mu|X)^-1;

o19 : RationalMap (birational map from hypersurface in PP^5 to X)

Computing associated K3 surfaces to rational fourfolds. Here let’s assume that X is a cubic fourfold
(resp., a GM fourfold) of discriminant d which contains a surface S that admits a congruence of curves, and
moreover we can produce a birational map µ|X : X 99K W as described in the previous subsection. Then
the inverse map of µ|X is defined by a linear system of hypersurfaces in W with points of multiplicity
e along a surface U ⊂ W , which turns out to be a projection of a K3 surface Ũ ⊂ Pg of degree d
and genus g = d/2 + 1. For details and precise results, see [10]; see also [6] and [12]. The function
associatedK3surface applied to the fourfold X returns this surface Ũ . As an example, we now
compute the K3 surface Ũ ⊂ P6 associated with the GM fourfolds X containing a σ3,1-plane.

i20 : U’ = associatedK3surface X;

o20 : ProjectiveVariety, K3 surface associated to X

i21 : describe U’

o21 = ambient:.............. PP^6
dim:.................. 2
codim:................ 4
degree:............... 10
generators:........... 2^6
purity:............... true
dim sing. l.:......... -1

The function associatedK3surface can be used to construct explicit general K3 surfaces of given
genus g. Currently, this can be done for g ∈ {6, 8, 11, 14, 20, 22}, and consequently for these values we
get an explicit unirationality of the moduli space Fg of polarized K3 surfaces of genus g. See [12] and
also [6; 10] for more explanations and references on this topic.

SUPPLEMENT. The online supplement contains version 2.7.1 of SpecialFanoFourfolds.

